NAG Library Routine Document

F11XNF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F11XNF computes a matrix-vector or conjugate transposed matrix-vector product involving a complex sparse non-Hermitian matrix stored in coordinate storage format.

2 Specification

```
SUBROUTINE F11XNF (TRANS, N, NNZ, A, IROW, ICOL, CHECK, X, Y, IFAIL)
INTEGER                 N, NNZ, IROW(NNZ), ICOL(NNZ), IFAIL
COMPLEX (KIND=nag_wp) A(NNZ), X(N), Y(N)
CHARACTER(1)            TRANS, CHECK
```

3 Description

F11XNF computes either the matrix-vector product $y = Ax$, or the conjugate transposed matrix-vector product $y = A^{\mathrm{H}}x$, according to the value of the argument TRANS, where A is a complex n by n sparse non-Hermitian matrix, of arbitrary sparsity pattern. The matrix A is stored in coordinate storage (CS) format (see Section 2.1.1 in the F11 Chapter Introduction). The array A stores all the nonzero elements of A, while arrays IROW and ICOL store the corresponding row and column indices respectively.

It is envisaged that a common use of F11XNF will be to compute the matrix-vector product required in the application of F11BSF to sparse complex linear systems. This is illustrated in Section 10 in F11DRF.

4 References

None.

5 Parameters

1: TRANS – CHARACTER(1) *Input*

On entry: specifies whether or not the matrix A is conjugate transposed.

TRANS = 'N'
$y = Ax$ is computed.

TRANS = 'T'
$y = A^{\mathrm{H}}x$ is computed.

Constraint: TRANS = 'N' or 'T'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 1.

3: NNZ – INTEGER *Input*

On entry: the number of nonzero elements in the matrix A.

Constraint: $1 \leq$ NNZ $\leq$ N^2.

4: A(NNZ) – COMPLEX (KIND=nag_wp) array *Input*

On entry: the nonzero elements in the matrix A, ordered by increasing row index, and by increasing column index within each row. Multiple entries for the same row and column indices are not permitted. The routine F11ZNF may be used to order the elements in this way.

5: IROW(NNZ) – INTEGER array *Input*

6: ICOL(NNZ) – INTEGER array *Input*

On entry: the row and column indices of the nonzero elements supplied in array A.

Constraints:

$1 \leq \text{IROW}(i) \leq \text{N}$ and $1 \leq \text{ICOL}(i) \leq \text{N}$, for $i = 1, 2, \ldots, \text{NNZ}$;
$\text{IROW}(i-1) < \text{IROW}(i)$ or $\text{IROW}(i-1) = \text{IROW}(i)$ and $\text{ICOL}(i-1) < \text{ICOL}(i)$, for $i = 2, 3, \ldots, \text{NNZ}$.

7: CHECK – CHARACTER(1) *Input*

On entry: specifies whether or not the CS representation of the matrix A, values of N, NNZ, IROW and ICOL should be checked.

CHECK = 'C'
Checks are carried on the values of N, NNZ, IROW and ICOL.

CHECK = 'N'
None of these checks are carried out.

See also Section 9.2.

Constraint: CHECK = 'C' or 'N'.

8: X(N) – COMPLEX (KIND=nag_wp) array *Input*

On entry: the vector x.

9: Y(N) – COMPLEX (KIND=nag_wp) array *Output*

On exit: the vector y.

10: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, TRANS $\neq$ 'N' or 'T',
or CHECK $\neq$ 'C' or 'N'.

IFAIL $= 2$

On entry, N < 1,
or NNZ < 1,
or NNZ $>$ N^2.

IFAIL $= 3$

On entry, the arrays IROW and ICOL fail to satisfy the following constraints:

$1 \leq \text{IROW}(i) \leq \text{N}$ and $1 \leq \text{ICOL}(i) \leq \text{N}$, for $i = 1, 2, \ldots, \text{NNZ}$;

$\text{IROW}(i-1) < \text{IROW}(i)$, or $\text{IROW}(i-1) = \text{IROW}(i)$ and $\text{ICOL}(i-1) < \text{ICOL}(i)$, for $i = 2, 3, \ldots, \text{NNZ}$.

Therefore a nonzero element has been supplied which does not lie within the matrix A, is out of order, or has duplicate row and column indices. Call F11ZNF to reorder and sum or remove duplicates.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The computed vector y satisfies the error bound:

$\|y - Ax\|_\infty \leq c(n)\epsilon\|A\|_\infty\|x\|_\infty$, if TRANS = 'N', or

$\|y - A^{\text{H}}x\|_\infty \leq c(n)\epsilon\|A^{\text{H}}\|_\infty\|x\|_\infty$, if TRANS = 'T',

where $c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

8 Parallelism and Performance

F11XNF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F11XNF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

9.1 Timing

The time taken for a call to F11XNF is proportional to NNZ.

9.2 Use of CHECK

It is expected that a common use of F11XNF will be to compute the matrix-vector product required in the application of F11BSF to sparse complex linear systems. In this situation F11XNF is likely to be called many times with the same matrix A. In the interests of both reliability and efficiency you are recommended to set CHECK $=$ 'C' for the first of such calls, and to set CHECK $=$ 'N' for all subsequent calls.

10 Example

This example reads in a complex sparse matrix A and a vector x. It then calls F11XNF to compute the matrix-vector product $y = Ax$ and the conjugate transposed matrix-vector product $y = A^{\mathrm{H}}x$.

10.1 Program Text

```
    Program f11xnfe

!     F11XNF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: f11xnf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, n, nnz
      Character (1)                    :: check, trans
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:), x(:), y(:)
      Integer, Allocatable             :: icol(:), irow(:)
!     .. Executable Statements ..
      Write (nout,*) 'F11XNF Example Program Results'
!     Skip heading in data file
      Read (nin,*)

!     Read order of matrix and number of non-zero entries

      Read (nin,*) n
      Read (nin,*) nnz

      Allocate (a(nnz),x(n),y(n),icol(nnz),irow(nnz))

!     Read the matrix A

      Do i = 1, nnz
        Read (nin,*) a(i), irow(i), icol(i)
      End Do

!     Read the vector x

      Read (nin,*) x(1:n)

!     Calculate matrix-vector product

      trans = 'N'
      check = 'C'

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call f11xnf(trans,n,nnz,a,irow,icol,check,x,y,ifail)

!     Output results

      Write (nout,*)
```

```
      Write (nout,*) ' Matrix-vector product'
      Write (nout,'(1X,''('',E16.4,'','',E16.4,'')'')') y(1:n)

!     Calculate conjugate transposed matrix-vector product

      trans = 'T'
      check = 'N'

      ifail = 0
      Call f11xnf(trans,n,nnz,a,irow,icol,check,x,y,ifail)

!     Output results

      Write (nout,*)
      Write (nout,*) ' Conjugate transposed matrix-vector product'
      Write (nout,'(1X,''('',E16.4,'','',E16.4,'')'')') y(1:n)

    End Program f11xnfe
```

10.2 Program Data

```
F11XNF Example Program Data
  5                       N
 11                       NNZ
  ( 2., 3.)   1    1
  ( 1.,-4.)   1    2
  ( 1., 0.)   2    3
  (-1.,-2.)   2    4
  ( 4., 1.)   3    1
  ( 0., 1.)   3    3
  ( 1., 3.)   3    5
  ( 0.,-1.)   4    4
  ( 2.,-6.)   4    5
  (-2., 0.)   5    2
  ( 3., 1.)   5    5     A(I), IROW(I), ICOL(I), I=1,...,NNZ
 ( 0.70, 0.21)
 ( 0.16,-0.43)
 ( 0.52, 0.97)
 ( 0.77, 0.00)
 ( 0.28,-0.64)           X(I), I=1,...,N
```

10.3 Program Results

```
 F11XNF Example Program Results

  Matrix-vector product
 (     -0.7900E+00,      0.1450E+01)
 (     -0.2500E+00,     -0.5700E+00)
 (      0.3820E+01,      0.2260E+01)
 (     -0.3280E+01,     -0.3730E+01)
 (      0.1160E+01,     -0.7800E+00)

  Conjugate transposed matrix-vector product
 (      0.5080E+01,      0.1680E+01)
 (     -0.7000E+00,      0.4290E+01)
 (      0.1130E+01,     -0.9500E+00)
 (      0.7000E+00,      0.1520E+01)
 (      0.5170E+01,      0.1830E+01)
```